

LES FONCTIONS DE RECHERCHE ET DE RÉFÉRENCE

RECHERCHEV

Définition

La fonction **RECHERCHEV** permet de **rechercher** une **valeur** dans la **1^{ère} colonne** d'un **tableau** pour **renvoyer** une **donnée** qui se situe dans la **même ligne** que la **valeur recherchée**.

NB : **souvent imbriquée** dans la fonction **SI**, **SIERREUR** ou **SI.NON.DISP**, elle peut être **remplacée** par **RECHERCHEX** (ou **EQUIVX** imbriquée dans **INDEX**) qui est **plus complète** et **plus facile** à utiliser.

La fonction **RECHERCHEH** est **similaire** à **RECHERCHEV**, sauf qu'elle **recherche** une **valeur** dans la **1^{ère} ligne** d'un **tableau** pour **renvoyer** une **donnée** qui se situe dans la **même colonne**, alors que **RECHERCHEV** **recherche** une **valeur** dans la **1^{ère} colonne** d'un **tableau** pour **renvoyer** une **donnée** qui se situe dans la **même ligne**.

Syntaxe

RECHERCHEV(**Valeur_cherchée**;Table_matrice;No_index_col;**[Valeur_proche]**)

Argument	Description
Valeur_cherchée (obligatoire)	Valeur à rechercher dans le tableau . Attention ! La valeur doit être située dans la 1^{ère} colonne du tableau , donc les données que tu veux extraire doivent être situées à droite de cette 1^{ère} colonne .
Table_matrice (obligatoire)	Tableau ou plage dans laquelle tu veux rechercher la valeur pour renvoyer une donnée qui se situe dans la même ligne .
No_index_col (obligatoire)	N° de colonne (position) qui contient la donnée que tu veux renvoyer . Par exemple, la 1^{ère} colonne du tableau correspond à la valeur 1 .
[Valeur_proche] (facultatif)	Type de correspondance souhaité : ⇒ FAUX <u>ou</u> 0 (correspondance exacte) : cherche la 1^{ère} valeur exactement égale à la valeur . ⇒ VRAI , 1 <u>ou</u> omis (correspondance approximative) : cherche la plus grande valeur inférieure ou égale à la valeur . Attention ! Les valeurs de la 1^{ère} colonne du tableau doivent être triées par ordre croissant .

Exemple 1 : correspondance exacte (FAUX ou 0)

Rappel : **RECHERCHEV**(Valeur_cherchée;Table_matrice;No_index_col;[Valeur_proche])

⇒ **FAUX** ou **0** (correspondance exacte) : cherche la 1^{ère} valeur exactement égale à la valeur.

Ce tableau liste les données de plusieurs salariés d'une entreprise. La valeur à rechercher est le matricule (cellule G1) dans la 1^{ère} colonne (colonne A) du tableau.

Tu veux renvoyer le prénom (cellule G2), le service (cellule G3), et le salaire brut mensuel (cellule G4) correspondant au matricule du salarié (cellule G1).

	A	B	C	D	E	F	G
1	Matricule	Prénom	Service	Salaire		Matricule	1004
2	1001	Hugo	RH	1 500 €		Prénom	Quentin (1) =RECHERCHEV(G\$1;A\$2:D\$5;2;FAUX)
3	1002	Julie	Compta.	1 800 €		Service	RH =RECHERCHEV(G\$1;A\$2:D\$5;3;FAUX)
4	1003	Léna	Compta.	2 100 €		Salaire	2 400 € =RECHERCHEV(G\$1;A\$2:D\$5;4;FAUX)
5	1004	Quentin	RH	2 400 €			

Remarques

- ⇒ (1) La valeur à rechercher ("1004" dans la cellule G1) est bien située dans la 1^{ère} colonne du tableau ("1005" dans la cellule A5). Les données que tu veux renvoyer sont donc à droite de cette 1^{ère} colonne.
- ⇒ (1) Le n° de colonne est égal à 2, car tu veux renvoyer le prénom du salarié ("Quentin") qui est situé dans la 2^e colonne du tableau (colonne B).
- ⇒ (1) Pourquoi les \$? Ils permettent de figer la ligne 1 de la valeur à rechercher et les lignes 2 et 5 du tableau pour étirer la formule jusqu'à G4 et modifier uniquement le n° de colonne (voir « Les références de cellule »).
- ⇒ La valeur à rechercher ("1004") n'existe pas dans la 1^{ère} colonne du tableau ("Matricule"). Cela signifie que RECHERCHEV ne trouve pas ce qu'elle est censée rechercher. Dans ce cas, la formule renverra l'erreur #N/A, sauf si tu utilises la fonction SI.NON.DISP ou SIERREUR :
 ="SI.NON.DISP(RECHERCHEV(G\$1;A\$2:D\$5;2;FAUX);"Mat. inconnu")" renvoie "Mat. inconnu" si "1004" (cellule G1) n'existe pas dans la 1^{ère} colonne du tableau (erreur #N/A).
 ="SIERREUR(RECHERCHEV(G\$1;A\$2:D\$5;2;FAUX);"Mat. inconnu")" renvoie "Mat. inconnu" si n'importe quelle erreur est renvoyée par la fonction RECHERCHEV.

Exemple 2 : correspondance approximative (VRAI, 1 ou omis)

Rappel : RECHERCHEV(Valeur_cherchée;Table_matrice;No_index_col;[Valeur_proche])

⇒ **VRAI, 1 ou omis (correspondance approximative)** : cherche la plus grande valeur inférieure ou égale à la valeur. **Attention !** Les valeurs de la 1^{ère} colonne du tableau doivent être triées par ordre croissant.

Ce tableau liste le chiffre d'affaires réalisé (CA HT) par une entreprise auprès de plusieurs clients. Les valeurs à rechercher sont les CA HT (B2 à B5) dans la 1^{ère} colonne (E2 à E4) du tableau.

Tu veux renvoyer le taux de remise applicable auprès de chaque client (colonne C) en fonction du CA HT qu'il t'a apporté :

- ⇒ De 0 € à 1 000 € : 0% de remise.
- ⇒ De 1 000 € à 10 000 € : 5% de remise.
- ⇒ 10 000 € et plus : 10% de remise.

	A	B	C	D	E	F
1	Client	CA HT	Taux de remise		(3) CA HT	Remise
2	Amazon	350 €	(1) 0% (2) =RECHERCHEV(B2;E\$2:F\$4;2;VRAI)		0 €	0%
3	Apple	2 500 €	5%		1 000 €	5%
4	LVMH	13 000 €	10%		10 000 €	10%
5	Google	6 000 €	5% =RECHERCHEV(B5;E\$2:F\$4;2;VRAI)			

Remarques

- ⇒ (1) RECHERCHEV cherche la plus grande valeur inférieure ou égale à la valeur pour renvoyer la donnée : "0" est la plus grande valeur <= "350" donc RECHERCHEV renvoie "0%".
- ⇒ (2) Les valeurs à rechercher ("CA HT" de B2 à B5) sont bien situées dans la 1^{ère} colonne du tableau ("CA HT" de E2 à E4). Les données que tu veux renvoyer sont donc à droite de cette 1^{ère} colonne.
- ⇒ (2) Le n° de colonne est égal à 2, car tu veux renvoyer le taux de remise ("0%") qui est situé dans la 2^e colonne du tableau (F2 à F4).
- ⇒ (2) Pourquoi les \$? Ils permettent de figer les lignes 2 et 4 du tableau pour étirer la formule jusqu'à C5 (voir « Les références de cellule »).
- ⇒ (3) Les CA HT (E2 à E4) sont bien triés par ordre croissant.